Kreisfunktion definieren

(%i1) `k(x,r):= sqrt(r^2-x^2);`

(%o1) $k(x,r):=\sqrt{r^2-x^2}$

Steigung des Halbkreises an einem Punkt u

(%i2) `Steigung = m: diff(k(u,r),u);`

(%o2) $Steigung=-\frac{u}{\sqrt{r^2-u^2}}$

y-Achsenabschnitt für Tangente ermitteln

(%i3) `solve(k(u,r)=m*u+b,b);`

(%o3) $[b=\frac{r^2}{\sqrt{r^2-u^2}}]$

Tangente zusammenbauen

(%i4) `m*x+b,%;`

(%o4) $\frac{r^2}{\sqrt{r^2-u^2}}-\frac{u\,x}{\sqrt{r^2-u^2}}$

Funktion für Tangente definieren t(x,r,u)

(%i5) `define(t(x,r,u),%);`

(%o5) $t(x,r,u):=\frac{r^2}{\sqrt{r^2-u^2}}-\frac{u\,x}{\sqrt{r^2-u^2}}$

Zeichnung des Viertelkreises für r=1 + Tangente

(%i6) `wxplot2d([t(x,1,0.4), k(x,1)], [x,0,1])$`

(%t6)



Nullstelle der Tangente berechnen

(%i7) solve(t(x,r,u)=0,x);

(%o7) $[\, x = \frac{r^2}{u} \,]$

Variable "Nullstelle" definieren

(%i8) Nullstelle: r^2/u;

(%o8) $\frac{r^2}{u}$

Fläche herleiten

(%i9) integrate(t(x,r,u),x,0,Nullstelle);

(%o9) $\frac{r^4}{2\, u \sqrt{r^2 - u^2}}$

Funktion A(r,u) definieren, die Flächeninhalt berechnen

(%i10) define(A(r,u),%);

(%o10) $A(r, u) := \frac{r^4}{2\, u \sqrt{r^2 - u^2}}$

Flächeninhalt in Abhängigkeit vom Punkt u (r=1)

(%i11) wxplot2d([A(1,u)], [u,0,1], [y,0,5])$

(%t11)



Ableitung der Flächenfunktion besorgen

(%i12) diff(A(r,u),u,1);

(%o12) $\frac{r^4}{2(r^2 - u^2)^{3/2}} - \frac{r^4}{2\, u^2 \sqrt{r^2 - u^2}}$

Wo liegt das Minimum? (aus Graph ersichtlich, dass Minimum und kein Max.)

```
(%i13) solve(%=0,u);
```

(%o13) $[u=-\frac{r}{\sqrt{2}}, u=\frac{r}{\sqrt{2}}]$

Fläche in Abhängigkeit von r

```
(%i14) A(r,r/sqrt(2));
```

(%o14) $\frac{r^3}{|r|}$

Funktion der Minimalen Fläche abhängig von r

```
(%i15) define(A[minimal](r),%);
```

(%o15) $A_{minimal}(r):=\frac{r^3}{|r|}$

Tangentengleichung für minimale Fläche

```
(%i16) t(x,r,r/sqrt(2))$
       define(t[minimal](x,r),%);
```

(%o17) $t_{minimal}(x,r):=\frac{\sqrt{2}\,r^2}{|r|}-\frac{r\,x}{|r|}$

einige Werte für r = 1,2,3

```
(%i18) Radius(1) = A[minimal](1);
       Radius(2) = A[minimal](2);
       Radius(3) = A[minimal](3);
```

(%o18) $\mathrm{Radius}(1)=1$

(%o19) $\mathrm{Radius}(2)=4$

(%o20) $\mathrm{Radius}(3)=9$

Tangentengleichungen dazu

```
(%i21) Tangente(1) = t[minimal](x,1);
       Tangente(2) = t[minimal](x,2);
       Tangente(3) = t[minimal](x,3);
```

(%o21) $\mathrm{Tangente}(1)=\sqrt{2}-x$

(%o22) $\mathrm{Tangente}(2)=2^{3/2}-x$

(%o23) $\mathrm{Tangente}(3)=3\sqrt{2}-x$

Besipiele grafisch dargestellt

```
(%i24) wxplot2d([k(x,1), k(x,2), k(x,3), t[minimal](x,1), t[minimal](x,2),
       t[minimal](x,3)], [x,0,5], [y,0,5])$
```

(%t24)



by >>genius<<